



中国科学技术大学  
University of Science and Technology of China

# 数字集成电路设计

## 第八章 电路仿真

白雪飞

中国科学技术大学微电子学院

- 引言
- SPICE简介
- 器件模型
- 器件表征
- 电路表征
- 互连线仿真



# 引言

## ■ 仿真 (Simulation)

- 将系统建立模型以表征其关键特性，并对这些关键特征作出近似模仿
- 模型表示系统自身，仿真表示系统的时序行为
- 不能完全依赖仿真而忽略手工分析
  - Simulation guided through insight gained from analysis

## ■ 仿真器 (Simulator)

- 工艺仿真器：预测工艺步骤中各个因素如何影响器件特性
  - Synopsys Sentaurus, Silvaco TCAD, ...
- 电路仿真器：预测表明性能和功耗的电压和电流
  - Cadence Spectre, Synopsys PrimeSim HSPICE, ...
- 逻辑仿真器：验证HDL设计是否具有正确的逻辑操作
  - Cadence Incisive/Xcelium, Synopsys VCS, Siemens ModelSim, Xilinx Vivado Simulator, ...
- 体系结构仿真器：预测体系结构的吞吐量和存储器访问模式等
  - gem5, zsim, ...

# SPICE简介

## ■ SPICE (Simulation Program with Integrated Circuit Emphasis)

- 最初由UC Berkeley于1970年代开发
- 求解描述电路元件的非线性微分方程
- 有众多的免费版本和商用版本
- 输入文件包含：网表、仿真选项、分析命令、器件模型
- 网表可以手工输入，也可以由EDA软件从原理图或版图生成

## ■ 主要电路仿真软件

- Cadence Spectre, PSpice
- Synopsys PrimeSim HSPICE, HSPICEplus
- Siemens Tanner T-Spice
- 华大九天 Empyrean ALPS (Accurate Large Capacity Parallel SPICE)
- Silvaco SmartSpice, ADI LTspice, ...
- 本章例题采用HSPICE，用于其他版本SPICE仿真时应加以修改

# SPICE电路元件和数量级前缀字母



## 常用SPICE电路元件

字母	元件
R	电阻
C	电容
L	电感
K	互感
V	独立电压源
I	独立电流源
M	MOS晶体管
D	二极管
Q	双极型晶体管
W	有损传输线
X	子电路
E	电压控制电压源
F	电流控制电流源
G	电压控制电流源
H	电流控制电压源

## SPICE数量级前缀

字母	单位前缀	数量级
A	atto	$10^{-18}$
F	femto	$10^{-15}$
P	pico	$10^{-12}$
N	nano	$10^{-9}$
U	micro	$10^{-6}$
M	milli	$10^{-3}$
K	kilo	$10^3$
X, MEG	mega	$10^6$
G	giga	$10^9$
T	tera	$10^{12}$

## ■ 直流源

■ `Vdd vdd gnd 2.5`


## ■ 分段线性源

■ `Vin in gnd PWL 0ps 0 100ps 0 150ps 1.0 1ns 1.0`

## ■ 梯形脉冲源

■ `Vck clk gnd PULSE 0 1 0ps 100ps 100ps 300ps 800ps`

`PULSE v1 v2 td tr tf pw per`



梯形脉冲波形



\* RC circuit response to rising input

\*-----

\* Parameters and models

\*-----

.option post

\*-----

\* Simulation netlist

\*-----

Vin in gnd pwl 0ps 0 100ps 0 150ps 1.0 1ns 1.0

R1 in out 2k

C1 out gnd 100f

\*-----



\* Stimulus

\*-----

.tran 20ps 1ns

.plot v(in) v(out)

.end



RC电路响应仿真

\* MOS I-V characteristics

\*-----

\* Parameters and models

\*-----

```
.include './models.sp'
```

```
.temp 70
```

```
.option post
```

\*-----

\* Simulation netlist

\*-----

```
Vgs g gnd 0
```

```
Vds d gnd 0
```

```
M1 d g gnd gnd NMOS W=100n L=50n
```



\*-----

\* Stimulus

\*-----

```
.dc Vds 0 1.0 0.05 SWEEP Vgs 0 1.0 0.2
```

```
.end
```



MOS管I-V特性仿真

```
* Unloaded unit inverter step response
```

```
*-----
```

```
* Parameters and models
```

```
*-----
```

```
.param SUPPLY=1.0
```

```
.option scale=25n
```

```
.include './models.sp'
```

```
.temp 70
```

```
.option post
```

```
*-----
```

```
* Simulation netlist
```

```
*-----
```

```
Vdd vdd gnd 'SUPPLY'
```

```
Vin a gnd PULSE 0 'SUPPLY' 25ps 0ps 0ps 35ps 80ps
```

```
M1 y a gnd gnd NMOS W=4 L=2 AS=20 PS=18 AD=20 PD=18
```

```
M2 y a vdd vdd PMOS W=8 L=2 AS=40 PS=26 AD=40 PD=26
```



```
*-----
```

```
* Stimulus
```

```
*-----
```

```
.tran 0.1ps 80ps
```

```
.end
```



反相器阶跃响应仿真

```
* Fanout-of-4 inverter delay
*-----
* Parameters and models
*-----

.param SUPPLY=1.0
.param H=4
.option scale=25n
.include './models.sp'
.temp 70
.option post

*-----
* Subcircuits
*-----

.global vdd gnd
.subckt inv a y N=4 P=8
M1 y a gnd gnd NMOS W='N' L=2
+ AS='N*5' PS='2*N+10' AD='N*5' PD='2*N+10'
M2 y a vdd vdd PMOS W='P' L=2
+ AS='P*5' PS='2*P+10' AD='P*5' PD='2*P+10'
.ends

*-----
* Simulation netlist
*-----

Vdd vdd gnd 'SUPPLY'
```


```
Vin a gnd PULSE 0 'SUPPLY' 0ps 20ps 20ps 120ps 280ps
X1 a b inv          * shape input waveform
X2 b c inv M='H'     * reshape input waveform
X3 c d inv M='H**2'  * device under test
X4 d e inv M='H**3'  * load
X5 e f inv M='H**4'  * load on load

*-----
* Stimulus
*-----

.tran 0.1ps 280ps
.measure tpdr          * rising prop delay
+ TRIG v(c) VAL='SUPPLY/2' FALL=1
+ TARG v(d) VAL='SUPPLY/2' RISE=1
.measure tpdf          * falling prop delay
+ TRIG v(c) VAL='SUPPLY/2' RISE=1
+ TARG v(d) VAL='SUPPLY/2' FALL=1
.measure tpd           * average prop delay
+ param='(tpdr+tpdf)/2'
.measure trise          * rise time
+ TRIG v(d) VAL='0.2*SUPPLY' RISE=1
+ TARG v(d) VAL='0.8*SUPPLY' RISE=1
.measure tfall          * fall time
+ TRIG v(d) VAL='0.8*SUPPLY' FALL=1
+ TARG v(d) VAL='0.2*SUPPLY' FALL=1
.end
```

## F04反相器延时仿真和测量

# FO4反相器延时仿真和测量结果



FO4反相器延时仿真和测量

```
* FO4 inverter P/N ratio optimization
*-----
* Parameters and models
*-----
.param SUPPLY=1.0
.option scale=25n
.include './models.sp'
.temp 70
.option post
*-----
* Subcircuits
*-----
.global vdd gnd
.subckt inv a y N=4 P=8
M1 y a gnd gnd NMOS W='N' L=2
+ AS='N*5' PS='2*N+10' AD='N*5' PD='2*N+10'
M2 y a vdd vdd PMOS W='P' L=2
+ AS='P*5' PS='2*P+10' AD='P*5' PD='2*P+10'
.ends
*-----
* Simulation netlist
*-----
Vdd vdd gnd 'SUPPLY'
Vin a gnd PULSE 0 'SUPPLY' 0ps 20ps 20ps 120ps 280ps
X1 a b inv P='P1' * shape input waveform
X2 b c inv P='P1' M=4 * reshape input waveform
```

```
X3 c d inv P='P1' M=16 * device under test
X4 d e inv P='P1' M=64 * load
X5 e f inv P='P1' M=256 * load on load
*-----
* Optimization setup
*-----
.param P1=optrange(8,4,16) * from 4 to 16, guess 8
.model optmod opt itropt=30 * max of 30 iterations
.measure bestratio param='P1/4' * best P/N ratio
*-----
* Stimulus
*-----
.tran 0.1ps 280ps SWEEP OPTIMIZE=optrange
+ RESULTS=diff MODEL=optmod
.measure tpdr * rising prop delay
+ TRIG v(c) VAL='SUPPLY/2' FALL=1
+ TARG v(d) VAL='SUPPLY/2' RISE=1
.measure tpdf * falling prop delay
+ TRIG v(c) VAL='SUPPLY/2' RISE=1
+ TARG v(d) VAL='SUPPLY/2' FALL=1
.measure tpd * average prop delay
+ param='(tpdr+tpdf)/2' goal=0
.measure diff * diff between delays
+ param='tpdr-tpdf' goal=0
.end
```

## FO4反相器P/N比优化

## ■ 仿真精度

- `.option accurate`
- 减小积分容差，提高仿真精度

## ■ 自动结束仿真

- `.option autostop`
- 完成所有 `.measure` 函数的计算时，提前结束瞬态分析，减少仿真时间

## ■ 仿真温度

- `.temp 0 70 125`
- 设置用于仿真的电路温度，以 $^{\circ}\text{C}$ 为单位

## ■ 直流工作点

- `.op`
- 输出静态工作点处的电压、电流、晶体管偏置条件

# 器件模型



## ■ Level 1模型 (Shichman-Hodges Model)

- 与Shockley模型接近，增加了沟道长度调制和体效应


## ■ Level 2和Level 3模型

- Level 2模型基于Grove-Frohnman公式，Level 3模型基于经验公式
- 增加速度饱和、迁移率退化、亚阈值导电、漏致势垒降低等效应

## ■ BSIM模型 (Berkeley Short-Channel IGFET Model)

- BSIM3v3 工业标准MOSFET模型，可用于深亚微米工艺
- BSIM4 基于阈值电压的Bulk MOSFET模型，用于130-nm~22/20-nm工艺节点
- BSIM-BULK 基于电荷的Bulk MOSFET模型
- BSIM-SOI 绝缘体上硅(SOI)器件模型
- BSIM-IMG 独立多栅(Independent Multi-Gate)器件模型
- BSIM-CMG 公共多栅(Common Multi-Gate)器件模型，即FinFET模型

## 扩散区面积和周长



	AS1 / AD2	PS1 / PD2	AD1 / AS2	PD1 / PS2
(a)	$w \times 5$	$2 \times w + 10$	$w \times 5$	$2 \times w + 10$
(b)	$w \times 5$	$2 \times w + 10$	$w \times 3$	$w + 6$
(c)	$w \times 5$	$2 \times w + 10$	$w \times 1.5$	$w + 3$

\* Shared contacted diffusion

M1 mid b bot gnd NMOS W='w' L=2 AS='w\*5' PS='2\*w+10' AD='w\*3' PD='w+6'

M2 top a mid gnd NMOS W='w' L=2 AS='w\*3' PS='w+6' AD='w\*5' PD='2\*w+10'





## 带有接触共享扩散区的晶体管SPICE模型

```
* Inverter step response across process corners
*-----
* Parameters and models
*-----
.option scale=25n
.param SUP=1.0  * Must set before calling .lib
.lib './opconditions.lib' TT
.option post
*-----
* Simulation netlist
*-----
Vdd vdd gnd 'SUPPLY'
Vin a gnd PULSE 0 'SUPPLY' 25ps 0ps 0ps 35ps 80ps
M1 y a gnd gnd NMOS W=4 L=2
+ AS=20 PS=18 AD=20 PD=18
M2 y a vdd vdd PMOS W=8 L=2
+ AS=40 PS=26 AD=40 PD=26
*-----
* Stimulus
*-----
.tran 0.1ps 80ps
.alter
.lib './opconditions.lib' FF
.alter
.lib './opconditions.lib' SS
.end
```





```
* opconditions.lib
*-----
* TT: Typical nMOS, pMOS, voltage, temp.
*-----
.lib TT
.temp 70
.param SUPPLY='SUP'
.include 'modelsTT.sp'
.endl TT
*-----
* SS: Slow nMOS, pMOS, low voltage, high temp.
*-----
.lib SS
.temp 125
.param SUPPLY='0.9 * SUP'
.include 'modelsSS.sp'
.endl SS
*-----
* FF: Fast nMOS, pMOS, high voltage, low temp.
*-----
.lib FF
.temp 0
.param SUPPLY='1.1 * SUP'
.include 'modelsFF.sp'
.endl FF
```

## 多设计角仿真





# 器件表征



65-nm工艺NMOS管I-V特性




65-nm工艺PMOS管I-V特性




180-nm工艺NMOS管I-V特性

# 阈值电压的提取方法



## 固定电流法



阈值电压定义为在指定漏极电流 $I_{crit}$ 时的栅电压



## 线性外推法、最大跨导法


从 $I_{ds}-V_{gs}$ 曲线最大斜率处外推得到栅电压





## MOS管阈值电压


阈值电压与晶体管的沟道长度、宽度、温度、工艺等因素有关  
电流并不在 $V_t$ 处骤降为零，且在纳米工艺截止器件仍十分显著



## 提取用于估算延时的等效栅电容的电路


仿真输入SPICE文件中将晶体管扩散区面积和周长设置为0

调整 $C_{\text{delay}}$ 直至X3和X6的延时相等为止，此时 $C_{\text{delay}}$ 等于X4的等效栅电容



## 提取用于估算功耗的等效栅电容的电路

$$C_{\text{eff-power}} = \frac{\int i_{\text{in}}(t) dt}{V_{\text{DD}}}$$




## 提取用于估算延时的等效寄生电容的电路

调整 $C_{\text{delay}}$ 直至X3和X4的延时相等为止，此时 $C_{\text{delay}}$ 等于M1漏极等效电容


## 提取扩散电容时使用的尺寸参数

	$W (\lambda)$	$AD (\lambda^2)$	$PD (\lambda)$	每微米等效电容 $C$
$C_{\text{gol}}$	1600	0	0	$C_{\text{delay}}/1600\lambda$ (per $\mu\text{m}$ )
$C_{\text{jb}}$	0	8000	0	$C_{\text{delay}}/8000\lambda^2$ (per $\mu\text{m}^2$ )
$C_{\text{jbsw}}$	0	0	1600	$C_{\text{delay}}/1600\lambda$ (per $\mu\text{m}$ )
$C_{\text{jbswg}}$	1600	0	1600	$C_{\text{delay}}/1600\lambda - C_{\text{gol}}$ (per $\mu\text{m}$ )
$C_d$ (单独接触)	1600	8000	3200	$C_{\text{delay}}/1600\lambda$ (per $\mu\text{m}$ gate width)
$C_d$ (共享接触)	3200	9600	3200	$C_{\text{delay}}/1600\lambda$ (per $\mu\text{m}$ gate width)
$C_d$ (合并不带接触)	3200	4800	3200	$C_{\text{delay}}/1600\lambda$ (per $\mu\text{m}$ gate width)




扇出为 $h$ 的反相器

$$t_{pdr} = \frac{R_p}{2} (3hC + 3C_d)$$



上升延时


$$t_{pdf} = R_n (3hC + 3C_d)$$



下降延时

## 扇出 $h$ 的反相器的RC延时模型

通过不同扇出时的延时以及栅电容 $C$ ，可以估算NMOS/PMOS的等效电阻 $R_n$ 和 $R_p$



## 提取等效串联电阻的电路

通过不同扇出时c至d的延时以及栅电容 $C$ ，可以估算NMOS/PMOS的等效串联电阻

# 电路表征

\* Inverter DC transfer characteristics

-----

\* Parameters and models

-----

```
.param SUPPLY=1.0
.option scale=25n
.include './models.sp'
.temp 70
.option post
```

-----

\* Simulation netlist

-----

```
Vdd vdd gnd 'SUPPLY'
```

```
Va a gnd 0
```

```
Vclk clk gnd 'SUPPLY'
```

\* Static Inverter

```
M1 y1 a gnd gnd NMOS W=16 L=2
```

```
M2 y1 a vdd vdd PMOS W=32 L=2
```

\* Dynamic Inverter

```
M3 y2 a gnd gnd NMOS W=16 L=2
```

```
M4 y2 clk vdd vdd PMOS W=16 L=2
```

```
M5 y2 z vdd vdd PMOS W=4 L=2
```

```
M6 z y2 gnd gnd NMOS W=4 L=2
```

```
M7 z y2 vdd vdd PMOS W=8 L=2
```

```
.ic V(y2)='SUPPLY'
```



-----

\* Stimulus


-----

```
.dc Va 0 1.0 0.01
```

```
.end
```




静态和动态反相器直流传输特性仿真



用于表征二输入与非门延时的电路

`.tran 1ps 1000ps SWEEP H 1 8 1`



65-nm工艺反相器和二输入与非门的  
延时与扇出的关系及逻辑努力

## ■ 电压源输出的电荷和能量

- `.measure charge INTEGRAL I(vdd) FROM=0ns TO=10ns`
- `.measure energy param='charge*SUPPLY'`

## ■ 电压源输出的瞬时功率和平均功率


- `.print P(vdd)`
- `.measure pwr AVG P(vdd) FROM=0ns TO=10ns`

## ■ 失配情况仿真

- 仅在不同设计角仿真无法体现晶体管的失配
- 仿真不同晶体管之间的PVT偏差

## ■ 蒙特卡洛仿真




- 器件模型必须包括参数的统计分布数据
- `.tran 1ps 1000ps SWEEP MONTE=30`



失配情况仿真



# 互连线仿真



有损多导体传输线模型和仿真波形

**本章结束**